

Freddy Brackx; Denis Constaes; Richard Delanghe; Herman Serras  
Clifford algebra with Reduce

In: Jarolím Bureš and Vladimír Souček (eds.): Proceedings of the Winter School "Geometry and Physics". Circolo Matematico di Palermo, Palermo, 1987. Rendiconti del Circolo Matematico di Palermo, Serie II, Supplemento No. 16. pp. [11]–19.

Persistent URL: <http://dml.cz/dmlcz/701405>

### Terms of use:

© Circolo Matematico di Palermo, 1987

Institute of Mathematics of the Czech Academy of Sciences provides access to digitized documents strictly for personal use. Each copy of any part of this document must contain these *Terms of use*.



This document has been digitized, optimized for electronic delivery and stamped with digital signature within the project *DML-CZ: The Czech Digital Mathematics Library* <http://dml.cz>

## CLIFFORD ALGEBRA WITH REDUCE

Freddy Brackx, Denis Constaes  
Richard Delanghe, Herman Serras

### 1. Introduction

Today certain computer software systems, commonly called Computer Algebra Systems or Symbolic Mathematical Systems, compute not only with numbers but also manipulate formulae and return answers in terms of symbols and algebraic or analytic expressions. They are characterized by exact computation rather than numerical approximation, and they can perform a large portion of the calculations usually done by hand.

The aim of the paper is to show briefly how algebraic calculations in Clifford algebra can be executed, independently of the dimension, by using the symbolic manipulation language REDUCE, version 3.2 [1], implemented on a VAX-750 computer..

### 2. Clifford algebra

2.1 Consider the  $n$ -dimensional real linear space  $\mathbb{R}^n$  with basis  $(e_1, e_2, \dots, e_n)$  and provided with the bilinear form

$$(e_i | e_j) = 0, i \neq j; (e_i | e_i) = -1, i=1, \dots, n.$$

If on the  $2^n$ -dimensional real linear space  $\mathbb{R}_n$  with basis

$$(e_A = e_{h_1} \dots e_{h_r}, A=(h_1, \dots, h_r) \in \mathcal{P}(1, \dots, n), 1 \leq h_1 < \dots < h_r \leq n),$$

a product is defined according to the rules

$$e_i \cdot e_i = (e_i | e_i) = -1, i=1, \dots, n$$

$$e_i e_j + e_j e_i = 2(e_i | e_j) = 0, i \neq j$$

then  $\mathbb{R}_n$  is turned into a linear associative but non-commutative algebra over  $\mathbb{R}$ , called the universal Clifford algebra over  $\mathbb{R}$ . For a detailed account of Clifford algebras we refer to e.g. [2]. In REDUCE the basic element  $e$  is declared to be a non-commutative operator with the following multiplication rules :

```

operator e;
noncom e;
e(0)**2:=e(0);
for all i such that i>0 let e(i)**2=-e(0);
for all i such that i>0 let e(i)*e(0)=e(i);
for all i such that i>0 let e(0)*e(i)=e(i);
for all i,j such that i>j and j>0 let e(i)*e(j)=-e(j)*e(i);

```

Of any clifford number  $x$  we can take its real and imaginary parts by:

```

procedure scapar(x);
  lterm(x,e(0));
procedure imapar(x);
  x=scapar(x);

```

EXAMPLE: for  $n=3$  the most general clifford number is given by:

$$X := E(3)*AA(5) + E(2)*E(3)*AA(7) + E(2)*AA(3) + E(1)*E(3)*AA(6) + \\ E(1)*E(2)*E(3)*AA(8) + E(1)*E(2)*AA(4) + E(1)*AA(2) + E(0)*AA(1);$$

```

scapar(x);
E(0)*AA(1)

```

```

imapar(x);
E(3)*AA(5) + E(2)*E(3)*AA(7) + E(2)*AA(3) + E(1)*E(3)*AA(6) + \\
E(1)*E(2)*E(3)*AA(8) + E(1)*E(2)*AA(4) + E(1)*AA(2)

```

2.2 A linear combination of the  $e_A$  with  $\text{card}(A)=p$  is called a  $p$ -vector; the  $p$ -vector part of a clifford number is often called its  $p$ -grade. The even subalgebra  $\mathbb{R}_n^+$  is the direct sum of the subspaces of the even multi-vectors.

Procedures for extracting the grades of a clifford number are:

```

procedure gradsplitt(expres,n);
  begin scalar y;
  y:=expres;
  y:=sub(e(0)=usergrad(0)*e(0),y);
  for i:=1:n do y:=sub(e(i)=usergrad(1)*e(i),y);
  return y
end;

```

```

procedure grade(expres,k,n);
  lcof(gradsplit(expres,n),usergrad(k));

```

EXAMPLE:

```

grade(x,1,3);
E(3)*AA(5) + E(2)*AA(3) + E(1)*AA(2)

grade(x,2,3);
E(2)*E(3)*AA(7) + E(1)*E(3)*AA(6) + E(1)*E(2)*AA(4)

```

2.3 On the Clifford algebra  $\mathbb{R}_n$  three important (anti-)involutions may be defined, similar to complex conjugation.

The main involution is the automorphism extending  $\hat{v} = -v$ , for any 1-vector  $v \in \mathbb{R}^n \subset \mathbb{R}_n$ ; the reversion is the anti-automorphism induced by the orthogonal involution  $\tilde{v} = v$  and obtained by reversing the order of the factors in each basic element  $e_A$ ; the conjugation is a combination of the two others defined by  $x^- = \tilde{x}^\wedge$ . The norm on  $\mathbb{R}_n$  defined by  $N(x) = x^-x$  is in general not a real number; so often its real part, which we call the clifford norm, is used. The corresponding procedures in REDUCE read as follows:

```

procedure bar(ex,n);
  begin scalar y;
    y:=gradsplit(ex,n);
    for i:=0 step 4 until n do
      y:=sub(usergrad(i)=1,usergrad(i+1)=-1,usergrad(i+2)=-1,
        usergrad(i+3)=1,y);
    return y
  end;

procedure norm(x,n);
  bar(x,n)*x;

procedure clifnorm(x,n);
  scapar(bar(x,n)*x)/e(0);

```

EXAMPLE:

**bar(x,3);**

```
- E(3)*AA(5) - E(2)*E(3)*AA(7) - E(2)*AA(3) - E(1)*E(3)*AA(6) +
E(1)*E(2)*E(3)*AA(8) - E(1)*E(2)*AA(4) - E(1)*AA(2) + E(0)*AA(1)
```

**norm(x,3);**

```
E(1)*E(2)*E(3)*(2*AA(8)*AA(1) - 2*AA(7)*AA(2) + 2*AA(6)*AA(3) -
2*AA(5)*AA(4)) + E(0)*(AA(8)2 + AA(7)2 + AA(6)2 + AA(5)2 + AA(4)2
+ AA(3)2 + AA(2)2 + AA(1)2)
```

**clifnorm(x,3);**

```
AA(8)2 + AA(7)2 + AA(6)2 + AA(5)2 + AA(4)2 + AA(3)2 + AA(2)2 +
AA(1)2
```

2.4 From a geometrical point of view a p-vector may be interpreted as a piece of an oriented p-dimensional subspace with a certain magnitude. This is the more clear when introducing the outer (or exterior) product and inner (or dot) product; for 1-vectors v and w we have :

$$vw = \frac{1}{2} (vw + wv) + \frac{1}{2} (vw - wv) = (v|w) + v \wedge w$$

$$= v \text{ dot } w + v \text{ ext } w ;$$

these definitions are then extended by induction. The outer product of a p-vector and a q-vector is a (p+q)-vector while their dot product is a |p-q|-vector.

We now give the procedure for the outer product, the one for the dot product being similar:

**procedure ext(a,b);**

```
begin scalar as,bs,u,s;
write "dimension n had to be read in! if not, cancel!";
as:=gradsplit(a,n);
bs:=gradsplit(b,n);
s:=0;
for i:=0:n do << u:=lcof(as,usergrad(i));
if u NEQ 0 then s:=s + (for j:=0:n sum
grade(u*lcof(bs,usergrad(j)),i+j,n)) >>;
return s
end;
```

The operations `ext` and `dot` are now declared to be infix operators in order to use them in the usual handwritten way:

```
infix ext;
infix dot;
precedence dot,*;
precedence ext,dot;
```

If  $y$  is another clifford number of the same dimension :

```
y := E(3)*BB(5) + E(2)*E(3)*BB(7) + E(2)*BB(3) + E(1)*E(3)*BB(6) +
E(1)* E(2)*E(3)*BB(8) + E(1)*E(2)*BB(4) + E(1)*BB(2) + E(0)*BB(1);
```

then we ask for:

```
x ext y;
E(3)*(AA(5)*BB(1) + AA(1)*BB(5)) + E(2)*E(3)*(AA(7)*BB(1) -
AA(5)*BB(3) + AA(3)*BB(5) + AA(1)*BB(7)) + E(2)*(AA(3)*BB(1) +
AA(1)*BB(3)) + E(1)*E(3)*(AA(6)*BB(1) - AA(5)*BB(2) + AA(2)*BB(5) +
AA(1)*BB(6)) + E(1)*E(2)*E(3)*(AA(8)*BB(1) + AA(7)*BB(2) -
AA(6)*BB(3) + AA(5)*BB(4) + AA(4)*BB(5) - AA(3)*BB(6) +
AA(2)*BB(7) + AA(1)*BB(8)) + E(1)* E(2)*(AA(4)*BB(1) - AA(3)*BB(2)
+ AA(2)*BB(3) + AA(1)*BB(4)) + E(1)*( AA(2)*BB(1) + AA(1)*BB(2)) +
E(0)*AA(1)*BB(1)
```

### 3. Matrix representations and inversion

3.1 In a classical way a real matrix representation of  $\mathbb{R}_n$  may be obtained as follows. Order the basic elements  $e_A$  in a certain way; here we choose

```
b(1) = e_0
b(2) = e_1
b(3) = e_2 , b(4) = e_1 e_2
b(5) = e_3 , b(6) = e_1 e_3 , b(7) = e_2 e_3 , b(8) = e_1 e_2 e_3
etc.
```

Then associate to each clifford number  $x$  the real  $2^n \times 2^n$ -matrix  $\Theta(x)$ , the entries of which are given by

$$\Theta(x)_{(K,L)} = [x.b(L)]_K, \quad K,L = 1, \dots, 2^n$$

where  $[\mu]_K$  denotes the  $b(K)$ -component of the clifford number  $\mu$ .

This matrix representation is an isomorphism between  $\mathbb{R}_n$  and  $\mathbb{R}^{2^n \times 2^n}$  and  $\Theta(x^-) = \Theta(x)^t$ , where  $t$  denotes the transpose.

The procedure executing this matrix representation runs as follows:

```

procedure matrep(x,n);
  begin scalar m,u,v;
    m:=2**n;
    u:=scapar(x);
    v:=gradsplit(x-u,n);
    matrix mt(m,m);
    for j:=1:m do mt(j,j):=u/e(0);
    for k:=1:(m-1) do << vv(k):=v*b(k);
      for j:=(k+1):m do mt(j,k):=coef(b(j),vv(k))>>;
    for k:=2:m do << for j:=1:(k-1) do mt(j,k):=mt(k,j) >>;
    for k:=2:m do << for j:=1:(k-1) do <<
      for i:=1 step 4 until n do
        mt(j,k):=sub(usergrad(i)=-1,usergrad(i+1)=-1,
          usergrad(i+2)=1,usergrad(i+3)=1,mt(j,k))>>>>;
    for k:=1:(m-1) do << for j:=(k+1):m do << for i:=1:n do
      mt(j,k):=sub(usergrad(i)=1,mt(j,k))>>>>;
    end;
  end;

```

3.2 Let  $\mathbb{C}_n = \mathbb{R}_n \otimes \mathbb{C}$  be the complexified clifford algebra. A matrix representation of  $\mathbb{C}_n$  which is of interest to physics is provided by the so-called gamma-matrices (see e.g. [3]). Let us briefly sketch how those gamma-matrices may be defined. Two cases have to be distinguished accordingly the parity of the dimension  $n$ .

**First case :  $n = 2m$**

Consider the isotropic subspace  $W_n = \text{span}_{\mathbb{C}}\{f_1, \dots, f_m\}$  of  $\mathbb{C}^n = \mathbb{R}^n \otimes \mathbb{C}$  where  $f_j = (e_{2j-1} + i e_{2j})/\sqrt{2}$ ,  $j = 1, \dots, m$ . Spinor space  $S_n$  is defined to be  $S_n = \Lambda W_n$ .

If  $W_n^*$  denotes the dual space of  $W_n$ , and  $\bar{W}_n = \text{span}_{\mathbb{C}}\{\bar{f}_1, \dots, \bar{f}_m\}$ , there is a natural map  $\sigma : \bar{W}_n \rightarrow W_n^* : \bar{w} \rightarrow \langle \cdot, \bar{w} \rangle$  and  $(\bar{f}_1, \dots, \bar{f}_m)$  is a dual basis of  $W_n$ . Hence there is a natural map  $\rho : \bar{W}_n \rightarrow \text{End}(S_n) : \rho(\bar{f}_j)\alpha = -\sqrt{2} \langle \bar{f}_j, \alpha \rangle$ , for any spinor  $\alpha \in S_n$ . Now  $\mathbb{C}^n = W_n \oplus \bar{W}_n$  and the gamma-matrices are defined by :

$$\gamma_{2j-1}^{(n)} = \rho(e_{2j-1}) = \frac{1}{\sqrt{2}} [\rho(f_j) + \rho(\bar{f}_j)]$$

$$\gamma_{2j}^{(n)} = \rho(e_{2j}) = \frac{1}{\sqrt{2}} \frac{1}{i} [\rho(f_j) - \rho(\bar{f}_j)]$$

Those gamma-matrices enjoy the following properties :

- (i)  $\gamma_i^{(n)} \gamma_j^{(n)} + \gamma_j^{(n)} \gamma_i^{(n)} = -2 \delta_{ij} I$  ,  $I = \text{identity matrix}$   
(ii)  $\gamma_i^{(n)}$  is anti-hermitian.

So there is an embedding of the clifford algebra  $\mathbb{C}_{2m}$  in  $\text{End}(S_n)$  and hence  $\rho(\mathbb{C}_{2m}) = \text{End}(S_n)$ .

Notice that when the spinor space  $S_n$  is split up into its even and odd parts :  $S_n = S_n^+ \oplus S_n^-$  then  $\gamma_i^{(n)} : S_n^+ \rightarrow S_n^-$ .

Second case :  $n = 2m-1$

Take the even sub-algebra  $\mathbb{C}_{2m}^+$  for one dimension higher; then  $\rho(a) : S_{2m}^+ \rightarrow S_{2m}^+$ ,  $a \in \mathbb{C}_{2m}^+$ , and  $\rho(\mathbb{C}_{2m}^+) = \text{End}(S_{2m}^+) \oplus \text{End}(S_{2m}^-)$ . By the REDUCE procedure "xtom" we now compute the gamma-matrices of the basic elements  $e_1, e_2, e_3, e_4$  in  $\mathbb{C}_4$  : [only the non-zero entries of the matrices are written down]

xtom(e(1),4);	xtom(e(3),4);
MAT(1,3) := (-1)	MAT(1,4) := (-1)
MAT(2,4) := 1	MAT(2,3) := (-1)
MAT(3,1) := 1	MAT(3,2) := 1
MAT(4,2) := (-1)	MAT(4,1) := 1
xtom(e(2),4);	xtom(e(4),4);
MAT(1,3) := - I	MAT(1,4) := - I
MAT(2,4) := - I	MAT(2,3) := I
MAT(3,1) := - I	MAT(3,2) := I
MAT(4,2) := - I	MAT(4,1) := - I

3.3 As is well known a clifford number has in general no inverse. With the above matrix representations, procedures are given for the inversion of a clifford number; if the number is non-invertible then REDUCE replies by "singular matrix". The procedures are based on the following scheme :

$$x \in \mathbb{C}_n \rightarrow \text{matrix } m(x) \rightarrow \text{inverse matrix } m^{-1}(x) \rightarrow x^{-1} \in \mathbb{C}_n$$

EXAMPLE :

Given the clifford number  $z$  :

$Z := E(3) + E(1)*E(2) + E(0);$



we compute its gamma-matrix and the inverse matrix :

<code>xtom(z,4);</code>	<code>1/xtom(z,4);</code>
<code>MAT(1,1) := I + 1</code>	<code>MAT(1,1) := ( - I + 3)/5</code>
<code>MAT(1,4) := (-1)</code>	<code>MAT(1,4) := 1/(2*I + 1)</code>
<code>MAT(2,2) := - I + 1</code>	<code>MAT(2,2) := (I + 3)/5</code>
<code>MAT(2,3) := (-1)</code>	<code>MAT(2,3) := ( - 1)/(2*I - 1)</code>
<code>MAT(3,2) := 1</code>	<code>MAT(3,2) := 1/(2*I - 1)</code>
<code>MAT(3,3) := - I + 1</code>	<code>MAT(3,3) := (I + 3)/5</code>
<code>MAT(4,1) := 1</code>	<code>MAT(4,1) := ( - 1)/(2*I + 1)</code>
<code>MAT(4,4) := I + 1</code>	<code>MAT(4,4) := ( - I + 3)/5</code>

The clifford number corresponding with that inverse matrix is then the inverse of the number  $z$  :

```
mtom(1/xtom(z,4),4);
( - E(3) + 2*E(1)*E(2)*E(3) - E(1)*E(2) + 3*E(0))/5
```

Finally a small control :

```
z*mtom(1/xtom(z,4),4);
E(0)
```

#### 4. Clifford group

The Clifford group  $\Gamma_n$  of the Clifford algebra  $\mathbb{R}_n$  consists of those invertible elements  $g$  in  $\mathbb{R}_n$  for which  $(g \cdot v \cdot g^{-1}) \in \mathbb{R}^n$  for any vector  $v \in \mathbb{R}^n$ ; it may also be defined by  $\Gamma_n = \{\text{products of non-zero vectors in } \mathbb{R}^n\}$ .

The subgroup of  $\Gamma_n$  consisting of the elements  $g$  with unit norm  $N(g) = g \cdot g = 1$ , is called the pingroup  $\text{Pin}(n)$ ; it gives a twofold covering of  $O(n)$ .

The even part of the pingroup, i.e.  $\text{Pin}(n) \cap \mathbb{R}_n^+$ , is the spingroup  $\text{Spin}(n)$  which provides a twofold covering of  $SO(n)$ .

By means of our REDUCE procedure "groupclas" we are able to determine whether or not a given clifford number is in one of the above groups.

## EXAMPLES:

```
z:=E(3) + E(1)*E(2) + E(0)
```

```
groupclas(z,3);
```

Your clifford number is NOT in the Clifford group !

```
z1 := - 14*E(3) - 17*E(2) - E(1)*E(2)*E(3) - 6*E(1)
```

```
groupclas(z1,3);
```

Your clifford number is in the Clifford group GAMMA(3) and

```
((- 14*E(3) - 17*E(2) - E(1)*E(2)*E(3) - 6*E(1))/(3*SQRT(58)))
```

is in the pin group PIN(3).

```
z2 := - E(2)*E(3) - E(1)*E(3) - E(1)*E(2) - E(0)
```

```
groupclas(z2,3);
```

Your clifford number is in the Clifford group GAMMAPLUS(3) and

```
((- E(2)*E(3) - E(1)*E(3) - E(1)*E(2) - E(0))/2
```

is in the spin group SPIN(3)

**Acknowledgment :** It is pleasure to thank Simonne Gutt, Michel Cahen and Frank Sommen for introducing us to the gamma-matrices and for the nice discussions we had on this subject.

**Notice:** the complete REDUCE programs for manipulating Clifford algebra can be obtained on simple request.

## REFERENCES

- [1] A.C. HEARN (ed.), REDUCE 3.2, The Rand Corporation, Santa Monica (April 1985).
- [2] I.R. PORTEOUS, Topological Geometry, 2nd. ed., Cambridge University Press (1981)
- [3] A. PRESSLEY and D. SEGAL, Loop groups, Clarendon Press, Oxford (1986)

"This paper is in final form and no version of it will be submitted for publication elsewhere".

F. BRACKX, H. SERRAS

Seminarie voor Wiskundige Analyse

Sint-Pietersnieuwstraat 39

B-9000 GENT (België)

D. CONSTALES, R. DELANGHE

Seminarie voor Algebra en

Functionaalanalyse

Galglaan 2

B-9000 GENT (België)